Verified compiler for a language with control effects

(Weryfikacja kompilatora dla języka programowania z efektami sterowania)

Paweł Dybiec

Praca magisterska

Promotor: Piotr Polesiuk

Uniwersytet Wrocławski
Wydział Matematyki i Informatyki
Instytut Informatyki

11 marca 2026

Abstract

We present a type system for a language with labeled delimited control operators. It utilizes polymorphic types and effects to allow expressions parametrized by effects. Our goal is to formally prove the correctness of such a language. Formalisation work was done in Agda programming language.

Tematem tej pracy jest stworzenie i zweryfikowanie języka z operatorami sterowania dla kontynuacji ograniczonych. Język ten używa polimorficznych typów aby pozwolić na wyrażenie obliczeń generycznych wobec pewnych efektów. Celem pracy jest zweryfikowanie poprawności definicji takiego języka. Formalizacja została przeprowadzona w języku programowania Agda.

Contents

Chapter 1

Introduction

Control operators have been present in languages for a long time. One notable example is the SCHEME programming language which originated in the 70s[1] and provides a *call/cc* operator. Such operators enable non-local jumps, which might be familiar to users of imperative languages with goto operator. But those operators unlike goto are handled by capturing continuation, which is stored as first-class value, able to be passed, called or dropped. Operator *call/cc* when called, catches whole rest of program in the continuation. When this continuation is called, it aborts rest of the program and replaces it with this continuation. Thus in example program `(call/cc (k . k (k 1)))` only inner k would be called.

Operators for delimited control introduced by Danvy, and Filinski[2] and independently by Felleisen[3] allow to control continuations more precisely. They usually come in pairs: for example, the operator *reset* delimits continuations caught by $shift$. This allows localized use of control effects, as the scope of the captured continuation much more visible, while still allowing non-local execution, which enables interesting computational effects such as emulating non-determinism, exceptions and failure. Usually such calculi jump from $shift$ to the closest enclosing *reset*. Reduction rule looks like this
`(reset E[(shift k expr)]) ⇒ (reset (( k . expr) ( v . (reset E[v]))))` therefore, for example `(reset (+ 1 (shift k (k (k 0)))))` would call k twice and result in value 2.

Danvy and Filinski[4] also describe similar operators $shift_0$ and $reset_0$. Their reduction does not introduce outermost *reset*_0, therefore subsequent calls to $shift_0$ remove innermost $reset_0$ one by one. Their example showcases this behaviour `reset_0(3 + reset_0(4 * shift_0 k in shift_0 c in E))`. The continuation k would be bound to `x . 4 •x` and c to `x. 3 + x`. In addition Dybvig et al.[5] show the formalisation of delimited control with multiple prompts where labels are used to match the corresponding operators.

The aim of this thesis is to formalise a typed version of such a calculus with poly-

morphic types (similar to Piróg, Polesiuk and Sieczkowski[6]), but for labeled delimited continuations (like Kazuki Ikemori, Youyou Cong, and Hidehiko Masuhara[7]). Compared to Ikemori et al., the main difference in our work is the use of first-class labels for delimited continuations, similarly to how Xie et al.[8] used them for algebraic effects. Balik and Polesiuk presented such a language[9] which currently serves as the core of Fram programming language. Formalising such a calculus could serve as a step in to demonstrate soundness of Fram's implementation.

The soundness of the presented language is proved using the standard technique of progress and preservation[10][11]. Formalisation work was done in Agda 2.8. The code archive also features nix derivation describing the exact environment to build both package and this paper to ensure reproducibility.

Chapter 2

Surface language

2.1 Syntax

We start with presenting the syntax of the surface language. We use kinds to distinguish between types and effects.

```
data Kind : Set where
  T : Kind
  E : Kind
```

Types and effects are defined as mutually recursive. A type is either a type variable `ttv`, an arrow, a `forallt` or a label `L eff at A / E`. We represent variables and type variables with de Bruijn indices. Effects datatype stores list of effects (represented as types) , it's used in arrow and forall constructors of type to mark effects of computation underneath. It will be also used in typing judgements to mark effects available in computation. Constructor `L` represents type of label expressions. It just stores an effect `eff` represented by a label and then type and effect of delimited computation labelled by this label.

```
module Types where
  Id : Set
  Id = ℕ
  data Type : Set
  Effects : Set
  data Type where
    ttv : Id → Type
    _-_>_ : Type → Effects → Type → Type
    forallt : Kind → Type → Effects → Type
    L_at_/_ : Type → Type → Effects → Type
  Effects = List Type
open Types
```

Constructors of expressions such as `var`, `lam`, `app`, `tlam`, and `tapp` behave as usual. `var` is a de Bruijn indexed variable, `lam` is a lambda abstraction, `app` is a function application, `tlam` is a type abstraction, storing kind of abstracted type and `tapp` is type application. Other constructors are responsible for continuations and labels. Constructor `new` binds labels used by $\texttt{shift}_0$ and $\texttt{reset}_0$ expressions to pair them up. Constructor $\texttt{shift}_0$ stores the label and expression (parametrized

by continuation) that replaces whole delimited computation. Last constructor of expressions is `reset₀ e (v . en) e'` which has 3 arguments, last one `e'` is label that is used to match up `reset₀`s with `shift₀`s. First argument `e` is delimited computation where `shift₀` can be used. So when `shift₀ k.es` is being evaluated, it aborts enclosing up to corresponding `reset₀`, replacing it with `es` is being evaluated with `k` bound to continuation representing evaluation context between `shift₀` and its `reset₀`. Second argument `x. en` is used when no corresponding shift aborts during evaluation of `e`. If `e` evaluates to `v` - then whole `reset₀` reduces to `en` where `x` is bound to `v`.

```
module Expr where
  data Expr : Set where
    var : ℕ → Expr
    lam : Expr → Expr
    app : Expr → Expr → Expr
    tlam : Kind → Expr → Expr
    tapp : Expr → Type → Expr
    new : Expr → Expr
    shift₀ : Expr → Expr → Expr
    reset₀ : Expr → Expr → Expr → Expr
open Expr
```

2.2 Typing judgements

Typing contexts are represented as lists of types, and contexts for type variables are represented as lists of kinds. Type or kind judgements for membership have Peano numbers structure.

```
module Typing where
  infixl 5 _,_
  data Context : Set where
    ∅ : Context
    _,_ : Context → Type → Context

  data TContext : Set where
    ∅ : TContext
    _,_ : TContext → Kind → TContext
```

Expressions are extrinsically typed, thus typing judgements are represented separately. Typing rules for `var`, `lam`, `app`, `tlam`, and `tapp` are defined mostly as usual. The noticeable difference is that value expressions are generic over effects they execute.

```
data _,_⊢_⦂_/_ : TContext → Context → Expr → Type → Effects → Set where
  ⊢var : ∀ {Γ Δ x A E}
    → Γ ∋ x ⦂ A
    ------------------------
    → Δ , Γ ⊢ var x ⦂ A / E

  ⊢lam : ∀ {Γ Δ e A B E F}
    → Δ , (Γ , A) ⊢ e ⦂ B / E
    ---------------------------------
    → Δ , Γ ⊢ lam e ⦂ A - E > B / F

  ⊢weak : ∀ {Γ Δ e A A' E E'}
    → Δ ⊢ A <t⦂ A'
    → Δ ⊢ E <⦂ E'
    → Δ , Γ ⊢ e ⦂ A / E
    --------------------
    → Δ , Γ ⊢ e ⦂ A' / E'
```

```
⊢app : ∀ {Γ Δ e1 e2 A B E}
  → Δ , Γ ⊢ e1 ꞉ A - E > B / E
  → Δ , Γ ⊢ e2 ꞉ A / E
  ----------------------------
  → Δ , Γ ⊢ app e1 e2 ꞉ B / E

⊢forall : ∀ {Γ Δ e k A E F}
  → (Δ , k) , Γ ⊢ e ꞉ TypeSubst.bump A / TypeSubst.bump' E
  -------------------------------------
  → Δ , Γ ⊢ tlam k e ꞉ forallt k A E / F

⊢tapp : ∀ {Γ Δ e k A T E}
  → Δ ⊢ T ꞉te
  → Δ , Γ ⊢ e ꞉ forallt k A E / E
  ---------------------------------------------------
  → Δ , Γ ⊢ tapp e T ꞉ A TypeSubst.[ T ] / (E TypeSubst.effs[t T ])
```

The `new` construct introduces a new type variable and a variable that represent respectively an effect and a label. The label type stores effect bound by `new` (here `ttv zero`) and `A1 / E2` representing a type and an effect of delimited computation.

```
⊢new : ∀ {Γ Δ e A A1 E E1}
  → Δ ⊢ A1 ꞉t
  → Δ ⊢ E1 ꞉effs
  → (Δ , Kind.E) , (Γ , (L ttv zero at A1 / E1)) ⊢ e ꞉ TypeSubst.bump A / TypeSubst.bump' E
  ----------------------
  → Δ , Γ ⊢ new e ꞉ A / E
```

Constructor `shift₀ e' (k . e)` uses only one effect `E` represented by label `e'`. For it to be a properly typed expression inside shift, the constructor binds extra variable `k`, which is where continuation will be plugged. Therefore the type of this expression `e` should the same type and effect as stored in label type. Its typing context should be expanded to account for contination, which type should be an arrow from `shift₀` type to type and effect of whole delimited computation. Since during evaluation `reset₀` will be removed, effect `E` it has introduced will not be present in direct subexpressions of `shift₀`.

```
⊢shift₀ : ∀ {Γ Δ e e' A A' E E'}
  → Δ ⊢ E ꞉e
  → Δ , Γ ⊢ e' ꞉ (L E at A' / E') / nil
  → Δ , (Γ , A - E' > A' ) ⊢ e ꞉ A' / E'
  ----------------------------------------
  → Δ , Γ ⊢ shift₀ e' e ꞉ A / (E :: nil)
```

The `reset₀ e (v. en) e'` constructor has three parameters. The first one is expression `e` that will have access to the effect, so its list of effects is expanded. The second one is continuation `v . en` that will handle the value `v` returned from the first argument `e`. And third is the label `e'`, whose effect is the same one as one introduced in `e`. Label type also stores the type and effects of the whole delimited computaiton.

```
⊢reset₀ : ∀ {Γ Δ e e' en A A' E E'}
  → Δ ⊢ E ꞉e
  → Δ , Γ ⊢ e ꞉ A / (E :: E')
  → Δ , (Γ , A) ⊢ en ꞉ A' / E'
  → Δ , Γ ⊢ e' ꞉ (L E at A' / E') / nil
  ------------------------------------
  → Δ , Γ ⊢ reset₀ e en e' ꞉ A' / E'
```

Now that the surface language and its typing rules have been defined for it, the next chapter will consider how it could be evaluated.

Chapter 3

Runtime language

3.1 Runtime language

Grammar defined in Chapter 2 does not produce any expression other than variable of label type, while `shift₀` and `reset₀` require the same labels. That translates into two options to define a reduction relation: going under `new` binders or introducing label expressions. We decided to go with second option and define another runtime expression language expanded by the notion of label values.

When `new` expression is evaluated, all occurrences of variables bound by it will have been replaced with the newly allocated label value. That means reduction relation would need to be defined using an allocator state.

To ensure type safety, we will expand type syntax to include allocated effects and add new effect context to typing judgments, which maps allocations to delimiter type and effect.

```
module Types_ where
  Id : Set
  Id = ℕ
  Label = ℕ
  data Type : Set
  Effects : Set
  data Type where
    ttv : Id → Type
    _-_>_ : Type → Effects → Type → Type
    forallt : Kind → Type → Effects → Type
    L_at_/_ : Type → Type → Effects → Type
    Effect : Label → Type
  Effects = List Type
open Types_
```

Most of the constructors in `RExpr` are the same as in `Expr`. Labels runtime values are represented by natural numbers.

```
module RuntimeExpr where
  data RExpr : Set where --runtime version
    var : ℕ → RExpr
    lam : RExpr → RExpr
    app : RExpr → RExpr → RExpr
    tlam : Kind → RExpr → RExpr
    tapp : RExpr → Type → RExpr
    new : RExpr → RExpr
```

```
    shift₀ : RExpr → RExpr → RExpr
    reset₀ : RExpr → RExpr → RExpr → RExpr
```

A separate term for the label stores the allocated label identifier.

```
label : Label → RExpr
```

Since the grammar of types has changed, we need to redefine contexts. This means almost all typing judgements need to be redefined. A new context is introduced that maps label values to delimiter type and effects. In this store typing technique described by Pierce[12],typing rules take maps from locations to types in order to ensure allocations' type safety. This context is represented by a finite vector of types and effects. As they are only used during evaluation, the types and effects have no free variables.

```
module Typing where
  open RuntimeExpr
  infixl 5 _,_
  data Context : Set where
    ∅ : Context
    _,_ : Context → Type → Context


  data TContext : Set where
    ∅ : TContext
    _,_ : TContext → Kind → TContext


  push : Kind → TContext → TContext
  push k Δ = Δ , k
  EContext = Σ[ n ∈ ℕ ] Data.Vec.Vec (Type × Effects) n
  pattern ∅' = 0 ,, Data.Vec.[]
```

Most typing judgements work as before. This paper does not discuss the ones that show up in surface language, as they simply pass extra context for effects throughout the rules.

```
private
  variable
    Γ : Context
    Δ : TContext
    Θ : EContext
    e e1 e2 : RExpr
    A B : Type
    E E' F : Effects
    k : Kind
infix 4 _;_;_⊢_∶_/_
data _;_;_⊢_∶_/_ : TContext → EContext → Context → RExpr → Type → Effects → Set where
```

Since the labels are represented by natural numbers, ensuring correct label type relies on a simple lookup in an appropriate context.

```
⊢label : ∀ {n }
  → Θ ∋l n ∶ A / E
  ---------------------------------------------
  → Δ ; Θ ; Γ ⊢ label n ∶ (L (Effect n) at A / E) / F
```

3.2 Runtime embedding

We defined embedding of the surface language in the runtime language one, and showed the proof that such embedding preserves typing judgements. Most of the proof is not displayed here, as it is repetitive, that is, it goes through almost all judgement types. The type of main proof is presented below.

```
runtime-types : ∀ {Δ Γ e T E}
  → Δ Types.Typing., Γ ⊢ e ꞉ T / E → (runtimeΔ Δ ; ∅' ; runtimeΓ Γ ⊢ (runtime e) ꞉ rt-t T / rt-t' E)
```

Chapter 4

Semantics

This chapter defines reduction relation and show its soundness.

4.1 Values

Only abstractions, type abstractions and labels are considered values. Since the values themselves do not perform any effects, they have a `nil` effect. But rules for all of them have a built-in weakening. We can use that to generalize their type and perform a substitution where any effect is expected.

```
data Value : RExpr → Set where
  vlam : ∀ { e } → Value (lam e)
  vLam : ∀ { k e } → Value (tlam k e)
  vlab : ∀ { n } → Value (label n)
gvalue : ∀ {Δ Θ Γ T E e} → (Value e) → (Δ ; Θ ; Γ ⊢ e ∶ T / E) → ∀ {F} → (Δ ; Θ ; Γ ⊢ e ∶ T / F)
gvalue vlam (⊢lam t) = ⊢lam t
gvalue vLam (⊢forall t) = ⊢forall t
gvalue vlab (⊢label x) = ⊢label x
gvalue {Δ} v (⊢weak x x₁ x₂) {F} = (⊢weak x ( <∶e-refl {Δ} {F}) (gvalue v x₂))
```

4.2 Frames

In this thesis, the term "frame" stands for what is usually evaluation context in literature. This term was sellected to be sufficiently different from typing context. The frame represents parts between $\texttt{reset}_0$s, or between $\texttt{reset}_0$ and $\texttt{shift}_0$. Frame type is parametrized by Θ Γ, that is, the typing context outside of frame, a T type of the hole. It's also indexed by whole frame type and effects, and the effects of the hole, . Frames are intrinsically typed, thus they also store type judgements of subexpressions. They are defined in such a way to reduce repetition. Otherwise we would need to introduce typing judgements for frames, and then prove type preservation for every operation such as plugging or composition.

```
data Frame (Θ : EContext) (Γ : Context) (T : Type) : Effects → Type → Effects → Set where
  fempty : ∀ {Eff}
  ------------------------------
```

```
    → Frame Θ Γ T Eff T Eff
  fapp₁ : ∀ {A B Eff E} → Frame Θ Γ T Eff (A - Eff > B) E
    → (e : RExpr) → { ∅ ; Θ ; Γ ⊢ e ꞉ A / Eff }
    -------------------------
    → Frame Θ Γ T Eff B E
  fapp₂ : ∀ {A B Eff E} → (e : RExpr) → { v : Value e}
    → { ∅ ; Θ ; Γ ⊢ e ꞉ ( A - Eff > B) / Eff }
    ------------------------------------------------------
    → Frame Θ Γ T Eff A E → Frame Θ Γ T Eff B E
  freset-label : ∀ {A E A' Eff C}
    → (e en : RExpr)
    → ∅ ; Θ ⊢ C ꞉e
    → ∅ ; Θ ; Γ  ⊢ e ꞉ A' / (C :: Eff)
    → ∅ ; Θ ; (Γ , A')  ⊢ en ꞉ A / Eff
    → Frame Θ Γ T nil (L C at A / Eff) E
    -------------------------
    → Frame Θ Γ T Eff A E
  fshift-label : ∀ {A E A' E' C}
    → (e : RExpr)
    → ∅ ; Θ ⊢ C ꞉e
    → ∅ ; Θ ; (Γ , A - E' > A' ) ⊢ e ꞉ A' / E'
    → Frame Θ Γ T nil (L C at A' / E') E
    ------------------------------------
    → Frame Θ Γ T (C :: nil) A E
```

Definition and types for frame plugging and composition:

```
plug : ∀ {Θ Γ T Eff A E}
  → Frame Θ Γ T Eff A E
  → (e : RExpr) → ∅ ; Θ ; Γ ⊢ e ꞉ T / E
  → Σ[ res ∈ RExpr ] (∅ ; Θ ; Γ ⊢ res ꞉ A / Eff)
_∘f_ : ∀ {Γ Θ Eff Eff' Eff'' A B C }
  → Frame Θ Γ B Eff A Eff'
  → Frame Θ Γ C Eff' B Eff''
  → Frame Θ Γ C Eff A Eff''
```

We can prove how plugging and composition relate. Plugging an expression into one frame and result of that into another frame results in the same value and type as plugging the very same expression into a composition of two frames.

```
∘f-lemma : ∀ {Γ Θ Eff Eff' Eff'' A B C }
  → (f1 : Frame Θ Γ B Eff A Eff' )
  → (f2 : Frame Θ Γ C Eff' B Eff'' )
  → (e : RExpr) → (t : ∅ ; Θ ; Γ ⊢ e ꞉ C / Eff'')
  → plug ( f1 ∘f f2) e t
  ≡ ((λ x → plug f1 (Data.Product.proj₁ x) (Data.Product.proj₂ x))(plug f2 e t))
```

A metaframe stores the whole evaluation context, split into frames separated by resets. Type parameters and indices work in the same way as in the frame. Unlike the frame, however the metaframe now stores `reset₀`s, so the lists of effects inside and outside the frame may differ. This means that their difference represents a list of effects handled by the frame. This observation can be used to prove that for well-typed expressions they decompose into metaframe and `shift₀` expression, and that metaframe should handle effect of the `shift₀`. This is true provided that typing context is empty. This metaframe then decomposes into two two further metaframes separated by `reset₀` which has the same label as above-mentioned `shift₀`.

```
data Metaframe (Θ : EContext) (Γ : Context) (T : Type) (Eff : Effects)
  : Type → Effects → Set where
  mfempty : Metaframe Θ Γ T Eff T Eff
  mfreset : ∀ { A B C Eff' }
    → (l : Label)
    → ∅ ; Θ ⊢ C ꞉e
    → ∅ ; Θ ; Γ ⊢ label l ꞉ (L C at B / Eff) / nil
    → (e : RExpr) → (∅ ; Θ ; Γ , A ⊢ e ꞉ B / Eff)
    → Metaframe Θ Γ T (C :: Eff) A Eff'
```

```
    ---------------------------------
    → Metaframe Θ Γ T Eff B Eff'
  mframe : ∀ {A Eff' B Eff'' }
    → Frame Θ     Γ A Eff B Eff'
    → Metaframe Θ Γ T Eff' A Eff''
    ----------------------------------------
    → Metaframe Θ Γ T Eff B Eff''
```

Similarly to frames, metaframes can be lifted and plugged into arbitrary contexts. They can also be composed with simple frames.

```
↑m : forall {Θ A B Eff Eff' Γ' Γ}
  → Metaframe Θ Γ       A Eff B Eff'
  → Metaframe Θ (Γ' ∔ Γ) A Eff B Eff'
mplug : ∀ {Θ Γ T Eff A E}
  → Metaframe Θ Γ T Eff A E
  → (e : RExpr) → ∅ ; Θ ; Γ ⊢ e ꞉ T / E
  → Σ[ res ∈ RExpr ] (∅ ; Θ ; Γ ⊢ res ꞉ A / Eff)
_∘m_ : ∀ {Γ Θ Eff Eff' Eff'' A B C }
  → Metaframe Θ Γ B Eff A Eff'
  → Metaframe Θ Γ C Eff' B Eff''
  → Metaframe Θ Γ C Eff A Eff''
_f∘m_ : ∀ {Γ Θ Eff Eff' Eff'' A B C }
  → Frame Θ Γ B Eff A Eff'
  → Metaframe Θ Γ C Eff' B Eff''
  → Metaframe Θ Γ C Eff A Eff''
```

4.3 Reduction

Since labels need to be allocated, the reduction relation is defined in terms of the expression and state.The state itself is just the next label to be allocated. As frames are intrinsically typed, we need to provide judgments representing expressions well-typedness.

```
infix 2 _↦_
State = EContext
data _↦_ : RExpr × State → RExpr × State → Set where
  {-
 ↦new : ∀ {e Δ Θ E T A1 E1}
  → {te : (Δ , Kind.E) ; Θ ;(∅ , L ttv zero at A1 / E1)  ⊢ e ꞉ E / T}
  → new e ,′ Θ  ↦  RExprSubstTyped._[_] e (label (Θ .proj₁)) {te = te} {te1 = {!!}} .proj₁  ,′ Θ --(Data.Vec._++_ Θ (  Data.Vec
 -}
  β-lam-app : ∀ {e V Θ Γ A B E }
    → {te : ∅ ; Θ ; (Γ , A) ⊢ e ꞉ B / E}
    → {tv : ∅ ; Θ ; Γ ⊢ V ꞉ A / E}
    → Value (lam e)
    → (v : Value V)
    → app (lam e) V ,′ Θ ↦ (proj₁ (RExprSubstTyped._[_] e V {te = te} {te1 = (gvalue v tv)})) ,′ Θ

  β-tlam-tapp : ∀ {k e T Θ}
    → Value (tlam k e)
    → tapp (tlam k e) T ,′ Θ ↦ e RExprSubst.e[t T ] ,′ Θ

  reset₀-vl : ∀ {V e' en Θ Γ A B E }
    → ( v : Value V)
    → {tv : ∅ ; Θ ; Γ ⊢ V ꞉ A / E}
    → {ten : ∅ ; Θ ; (Γ , A) ⊢ en ꞉ B / E}
    → reset₀ V en e' ,′ Θ ↦ (RExprSubstTyped._[_] en V {te = ten} {te1 = (gvalue v tv)} .proj₁) ,′ Θ

{-
 reset₀-k : ∀ {es en e' e s n Θ A T Eff Eff' B A' E' ls lr}
  → { f : Metaframe Θ ∅ T Eff A Eff'  }
  → ∀ {cont-type}
  → Value (e')
  --shift
  → {ts : ∅ ; Θ ; ∅ ⊢ (shift₀ e' es) ꞉ T / Eff'}
  → {tes : ∅ ; Θ ; (∅ , T - Eff' > B) ⊢ es ꞉ B / Eff' }
  → {tls : ∅ ; Θ ; ∅ ⊢ e' ꞉ (L ttv ls at B / Eff') /  nil }
  → {tlvs : ∅ ; Θ  ⊢  ttv lr ꞉e }
```

```
  --reset
  → {te : Δ ; ∅ ⊢ e ꞉ A' / (ttv lr :: E') }
  → {ten : Δ ; ∅ , A' ⊢ en ꞉ T / Eff }
  → {tlr : Δ ; ∅ ⊢ e' ꞉ (L ttv lr at T / Eff) /  nil }
  → {tlvr : Δ  ⊢  ttv lr ꞉e }
  → (proj₁ (mplug  f (shift₀ e' es) ts)) ≡ e
  → reset₀ e en e' ,′ s
    ↦ RExprSubstTyped._[_] es
      (lam (reset₀ (proj₁ (mplug (↑m {Γ' = ∅ , T} f) (var 0) (⊢var Z))) en e'))
      {te = tes} {te1 = gvalue {E = Eff} vlam cont-type}
      .proj₁  ,′ s
-}
```

Since the simple reduction above is defined directly on redexes, we introduce -→ that represents the reduction within the metaframe. As only whole typed expressions are considered, an empty context is used instead of Γ.

```
{-
infix 2 _-→_
data _-→_ : RExpr × State → RExpr × State → Set where
  -→frame : ∀ {e1 e1' e2 e2' s s' n Δ Δ' A T Eff Eff' t1 t2} → (f : Metaframe Δ ∅ T Eff A Eff' Δ' n)
    → e1' ,′ s ↦ e2' ,′ s'
    → Data.Product.proj₁ (mplug f e1' t1) ≡ e1
    → Data.Product.proj₁ (mplug f e2' t2) ≡ e2
    →  (e1 ,′ s) -→ (e2 ,′ s')
-}
```

4.4 Progress

```
{-
data Decompose : State → (e : RExpr)  → Set where
  de-simpl-redex : ∀ {e e2 s s' n Δ Δ' A T Eff Eff'}
    → (f : Metaframe Δ ∅ T Eff A Eff' Δ' n)
    → (e ,′ s) -→ (e2 ,′ s')
    → (t : Δ ; ∅ ⊢ e ꞉ A / Eff)
    → Decompose s e
  de-shift : ∀ {s Δ Δ' T Eff A n Eff' es es' e l t}
    → (f : Metaframe Δ ∅ T Eff A Eff' Δ' n)
    →  shift₀ (label l) es' ≡ es
    → Data.Product.proj₁ (mplug f es t) ≡ e
    → (t : Δ ; ∅ ⊢ e ꞉ A / Eff)
    → (ts : Δ ; ∅ ⊢ es ꞉ T / Eff')
    → Decompose s e
  de-val : ∀ {Δ Eff A s e} → { t : Δ ; ∅ ⊢ e ꞉ A / Eff}
    → Value e
    → Decompose s e
data Progress : State → RExpr → Set where
  done : ∀ {e s} →  Value e → Progress s e
  step : ∀ {e1 s1 e2 s2}
    → ( e1 ,′ s1 ) -→ ( e2 ,′ s2)
    → Progress s1 e1
-}
```

Proof of progress would have a type of `progress : ∀ {A Δ Effs} → (s : State) → (e : RExpr) → (t : Δ ; ∅ ⊢ e ꞉ A / Effs) → Progress s e`. In such proof We would use auxiliary struct `Decompose` which builder would walk down well typed expression recursively until it has reached either value, simple reduction (app, tapp, new), or shift, and return it with the surrounding metaframe.

In case of shift, such a metaframe by construction should have an effect handler that has the same effect as shift. So we can construct `rest₀-k` and surrounding metaframe. Other cases would either be immediate value, or simple reduction in context.

4.4.1 Preservation

Current definition of reduction relation would yield proof of preservation immediately if progress was given since, frames and expressions are well typed, and reduction relation requires proofs to plug into metaframes or substitution.

Chapter 5

Discussion/Closing thoughts

This paper presents a language with labeled delimited continuations together with a partial language formalisation. Two term languages are defined, alongside typing judgements for them and translations between them. Typing is preserved when embedding surface language in runtime language. Intrinsically typed frames and metaframes allow to define the reduction relation and show how it preserves types. Since both frames and metaframes are typed, we statically know which effects are handled by the frames. This formalisation could be further expanded to prove the language correctness and its translations.

Bibliography

[1] Guy Lewis Steele, Gerald Jay Sussman. The Revised Report on SCHEME: A Dialect of LISP. In MIT Artificial Intelligence Memo 452, 1978. URL: https://dspace.mit.edu/handle/1721.1/6283.

[2] Olivier Danvy, Andrzej Filinski. Abstracting control. In LISP and Functional Programming, pages 151-160, 1990. URL: https://dl.acm.org/doi/10.1145/91556.91622

[3] Matthias Felleisen. The theory and practice of first-class prompts. In POPL '88: Proceedings of the 15th ACM SIGPLAN-SIGACT symposium on Principles of programming languages, pages 180-190, 1988. URL: https://dl.acm.org/doi/10.1145/73560.73576

[4] Olivier Danvy, Andrzej Filinski. A functional abstraction of typed contexts. In DIKU Rapport 89/12, DIKU, Computer Science Department, University of Copenhagen, Copenhagen, Denmark. July 1989.

[5] R. Kent Dybvig, Simon Peyton Jones, and Amr Sabry. A Monadic Framework for Delimited Continuations. In Journal of Functional Programming 17, no. 6 (2007): 687–730. 2007. URL: https://doi.org/10.1017/S0956796807006259

[6] Maciej Piróg, Piotr Polesiuk, Filip Sieczkowski. Typed Equivalence of Effect Handlers and Delimited Control. In 4th International Conference on Formal Structures for Computation and Deduction (FSCD 2019), pages 30:1-30:16. 2019. URL: https://doi.org/10.4230/LIPIcs.FSCD.2019.30

[7] Kazuki Ikemori, Youyou Cong, and Hidehiko Masuhara. Typed Equivalence of Labeled Effect Handlers and Labeled Delimited Control Operators. In International Symposium on Principles and Practice of Declarative Programming (PPDP 2023), October 22–23, 2023, Lisboa, Portugal. ACM, New York, NY, USA, 13 pages. 2023. URL: https://doi.org/10.1145/3610612.3610616

[8] Ningning Xie, Youyou Cong, Kazuki Ikemori, and Daan Leijen. First-Class Names for Effect Handlers. Proc. ACM Program. Lang. 6, OOPSLA2, Article 126 (October 2022), 30 pages. 2022. URL: https://doi.org/10.1145/3563289

[9] Patrycja Balik, and Piotr Polesiuk. Unpublished notes. 2024

[10] A.K. Wright, M. Felleisen. A Syntactic Approach to Type Soundness. In Information and Computation, Volume 115, Issue 1, Pages 38-94, 1994. URL: https://doi.org/10.1006/inco.1994.1093

[11] Robert Harper. Practical Foundations for Programming Languages. Cambridge University Press, New York, NY, USA, 2nd edition, 2016

[12] Benjamin C. Pierce. Types and programming languages. MIT press, 2002.